Using Maple in Calculus: Part 4

IMPLICIT DIFFERENTIATION

Finding implicit derivatives in Maple is very easy. Since we do not know what y is as a function of x, we can't *define* $f(x)$, but we can give a *name* to the equation using the command **`f:=`** followed by the equation in question.

Example: Name the equation $xy^3 + 3x = 2y - 5$.

In Maple, we are not restricted to a single letter for a name. We can use anything we like for the name of an equation, multiple letters and even entire words. An often-used name is **eqn** (for obvious reasons).

```
eqn:=x*y^3+3*x=2*y-5;
```

Don't forget to use the **`*`** for multiplication and the **`^`** for exponentiation, as well as a semicolon (**`;`**) at the end of the line.

Also note that there are two equal signs in the line, one for the equation itself and another for naming the equation.

Once we have named an equation, we can then find the implicit derivative by issuing the following command:

```
implicitdiff(name,y,x);
```

where **`name`** is the name we assigned to the equation, **`y`** is the variable that we want the implicit derivative of, and **`x`** is the variable that we are differentiating with respect to.

Because we will probably want to use the implicit derivative for something later on (such as finding the equation of a tangent line), it would be a good idea to name it as well. Popular choices are **`name1`** (for the first derivative of **`name`**) and **`nameprime`**, where **`name`** is the name we gave to the original equation. So we use the following more complete command:

```
nameprime:=implicitdiff(name,y,x);
```

Example: Using the same example as above, find the implicit derivative of y with respect to x.

```
eqnprime:=implicitdiff(eqn,y,x);
```

and Maple returns:

$$eqnprime := -\frac{y^3 + 3}{3xy^2 - 2}$$

Example: Find y' for $x\cos y + x^2 = \ln x$

First we name the equation, this time calling it **`mess`**:

```
mess:=x*cos(y)+x^2=ln(x);
```

Next we find the implicit derivative, calling it **`messprime`**:

```
messprime:=implicitdiff(mess,y,x);
```

SLOPE OF THE TANGENT LINE

Now that we know how to find y', we have a *formula* for the slope of the tangent line for any point on the curve, and we can therefore use it to find the slope for the tangent line at a specific point (x, y) on the curve.

Example: Find the slope of the tangent line to the curve given by $3x^2 - 4y^2 = 12$ when $x = 3$.

First, we name the equation:

```
eqn:=3*x^2-4*y^2=12;
```

Second, we find the implicit derivative:

```
eqnprime:=implicitdiff(eqn,y,x);
```

Third, we find the *y*-value that goes with $x = 3$. We do this with the **`subs`** command, followed by the **`solve`** command, both applied to the original equation:

```
subs(x=3,eqn);

solve(%,y);
```

Recall that the **%** sign is used to refer to the result on the immediately-preceding line.

At this point, Maple returns ***two*** values:

$$\frac{1}{2}\sqrt{15}, -\frac{1}{2}\sqrt{15}$$

We should not be all that surprised by this, since we probably do not have a *function* of *x*, so it is entirely possible that for one value of *x*, we might get two (or more) values of *y*. Since the question did not specify which value of *y* to use, we will find the slope of the tangent line for both of them.

Fourth, find the value of the derivative, **eqnprime**, when $x = 3$ and $y = \frac{1}{2}\sqrt{15}$, and then again when $x = 3$ and $y = -\frac{1}{2}\sqrt{15}$. We do this once again with the **subs** command (twice) applied to the derivative **eqnprime**:

```
subs(x=3,y=1/2*sqrt(15),eqnprime);
```

and

```
subs(x=3,y=-1/2*sqrt(15),eqnprime);
```

Maple returns values of $\frac{3}{10}\sqrt{15}$ and $-\frac{3}{10}\sqrt{15}$ for the slopes of the two tangent lines when $x = 3$.

GRAPHING IMPLICITLY-DEFINED FUNCTIONS

Before we can graph an equation that implicitly defines *y* as a function of *x*, we must first load some extra instructions into Maple. We do this with the following command:

```
with(plots);
```

Maple will return multiple lines of text showing you all the new commands that have been loaded into the program (many more commands than we will ever need!).

Once we have issued the **with(plots)** command, we are ready to graph using the **implicitplot** command:

```
implicitplot(name,x=a..b,y=c..d);
```

where **name** is the name of the equation, **a** is the left bound, **b** is the right bound, **c** is the lower bound, and **d** is the upper bound.

Example: Graph $xy - x^3 = 4 - y^2$.

Since no bounds are given, we have to set them ourselves. It may take some experimenting to find values for the bounds that "work" (that is, show a good portion of the graph). This process is similar to setting the window on a graphing calculator. Start with large values and make them smaller if possible.

In this example, it turns out that –6 and 6 are good choices for both the *x* bounds and the *y* bounds, but we could use larger or smaller values if we wanted to.

First, as always, we must name the equation:

```
eqn:=x*y-x^3=4-y^2;
```

Next, we graph the equation, using the name **eqn** and the values we discussed earlier:

```
implicitplot(eqn,x=-6..6,y=-6..6);
```

Maple gives us the following result, which we immediately recognize as not being a function:



GRAPHING A TANGENT LINE

Now that we know how to graph almost anything, we can easily add a tangent line to the graph. To do so, we will have to find the equation of the tangent line, but Maple can do most of the work for us. In the example immediately above, suppose we wanted to draw the tangent line when $x = 2$, for example. We see that there are really two such tangent lines, one for a point above the *x*-axis and the other below. Let's just add the one below.

As we know, the equation of the tangent line is the following:

$$y - y_1 = m_{\tan}(x - x_1)$$

where (x_1, y_1) is the specific point on the graph and $m_{\tan}$ is the slope of the tangent line at that point.

Example: Find and graph the tangent line to the curve $xy - x^3 = 4 - y^2$ when $x = 2$, choosing the point of tangency below the *x*-axis.

We begin by naming the equation:

```
eqn:= x*y-x^3=4-y^2
```

Next, we find the *y*-value when $x = 2$:

```
subs(x=2,eqn);
```

Now we have to find the *y*-value for the point in question:

```
solve(%,y);
```

Maple gives us the two values $-1 + \sqrt{13}, -1 - \sqrt{13}$, and we choose the one that is negative (below the *x*-axis as requested).

Before we can find the slope of the tangent line, we need to find the derivative (which we know is a formula for the slope of the tangent line):

```
eqnprime:=implicitdiff(eqn,y,x);
```

Maple returns $\dfrac{-y + 3x^2}{x + 2y}$.

Next we substitute our chosen value of *x* ($-1 - \sqrt{13}$) into this last result:

```
subs(x=2,y=-1-sqrt(13),eqnprime);
```

Maple returns $-\dfrac{1}{26}\left(13 + \sqrt{13}\right)\sqrt{13}$, not a pretty answer, but an answer nonetheless!

We can tell Maple to give us the decimal approximation of this answer by issuing the `evalf` command:

```
evalf(%);
```

Maple returns `-2.302775638`, which we will round off to `-2.3` for future calculations.

While we're at it, we might as well also find a decimal approximation for the *y*-value of $-1 - \sqrt{13}$:

```
evalf(-1-sqrt(13));
```

Maple returns `-4.605551275`, which we will round off to `-4.6`.

We are now ready to put the pieces together to find the equation of the tangent line. Recapping, we have $x = 2$, $y = -4.6$, and $m_{\tan} = -2.3$.
So our point-slope equation becomes:

$$y - y_1 = m_{\tan}(x - x_1)$$
$$y - \left(-4.6\right) = -2.3(x - 2)$$
$$y = -2.3x$$

We are now ready to graph both the original equation and the tangent line on the same set of axes. First we have to name our tangent line so we can refer to it later. We'll call it **`t`**, but we can choose any name we like.

```
t:=-2.3*x;
```

Because we want to do an "implicit" plot as well as an "ordinary" plot simultaneously, we have to name each plot before we can actually display them.

We'll use (1) **`curve`** for the name of the plot of the original graph and (2) **`tanline`** for the name of the plot of the tangent line. Finally, we'll instruct Maple to (3) **`display`** both of them at once on the same set of axes. We must to do all three of these things on the same command line, separating the three parts with colons (**`:`**). A colon simply tells Maple that you are issuing multiple commands at the same time. Since such a line can get quite long, it will automatically wrap around to the next line if necessary.

```
curve:=implicitplot(eqn,x=-6..6,y=-6..6):
tanline:=plot(t,x=-6..6,y=-6..6):display(curve,tanline);
```